
OVERVIEW





The presentation illustrates the technique of using special purpose VPU "Loop Tables" to create VPU resident tagnames, which operationally behave and appear identical to Unitec loops.  The major difference is that the user has the flexibility (and responsibility) to calculate his own feedback for the loop's process value, rather than using an automatic update from a Unitec I/O point.  The discussion focuses on the user configuration requirements to support trending, process value updating, and tagline access for this "VPU resident" loop type.





INTODUCTION





In the majority of Vision system installations, "Real Time Trending" is relied upon for quick, convenient display of recent process behavior, especially during upset conditions.  The Vision system implements this trend function by the use of trend devices, which perform data collection and buffering on the Unitec processors.  Each appropriate Unitec processors (AIP, CCP, HLA, LLA, PIDP, PIP) has one trend device per analog (or pulse) input channel.





However, there are often situations where a user needs to track additional data which is not available through this "normal" Unitec based trending.  This could be trending a Set Point or Process Output from a board (where all existing trend devices are already used for tracking Process Values) or to trend a calculated value such as delta pressure or a flow totalization for a batch charge meter.  It is important that additional trending should be transparent to the user, such that access methods and intermixing on trend displays with Unitec trending not have any noticeable differences.





Additionally, there are sometimes situations where it would be helpful to create supervisory or special purpose loops to allow setpoint entry access on the tagline.  This would allow operator entry to VPU resident controller setpoints or other operator entered limits/targets using the convenient and familiar tagline entry.  As with the trends, the transparency between this and Unitec loop types during tagline operation is a benefit to operating personnel.





Both of these needs occurred in conjunction with a batch charge meter implementation.  The discussion describes the software used to implement this batch charge meter using certain special purpose VPU loop and data table arrays to create the VPU resident loop which optionally supports trending.  The discussion will also show the code used to update the process value feedback and status (AUTO/MANL/ALRM) to complete the apparent functionality match with Unitec loops.  





The example also makes use of the operator entered setpoint and Auto/Manual mode changes to control the charge.  This VPU code makes outputs to Unitec program devices to effect the control of the charging logic.


�
UNITEC  LOOP/TAG  TYPES





Many of you are familiar with the loop tables and data tables from your work with Unitec based I/O points.  As a reference, a recent Vision Perspectives newsletter contains a reprint of a past Venus paper that discusses the topic.  





The primary purpose of the Vision system "tagname" function is to provide a structure for process data communications between I/O, the applications software and the operator interface.  The first illustration shows a loop supporting a Unitec  Controller Device  which "contains" the I/O, both the analog input and process output.  Additionally, there is a Trend Device, which is configured to monitor one measurement per device (this can be anything PV, SP, PO, but is usually set to PV).  





Notice that the "Loop Table Array" is the master unifying element that ties together all of the other elements (using address pointers).  This is called "encapsulating", where several elements are bound together to make something bigger, in this case making a "Tag" or "Loop" (depending on your naming preference).  Both Unitec devices also have pointers back to the Loop Table, which in turn has pointers to specify storing the process data reported from the Unitec I/O into the associated Data Table Array or trend buffer.


��
VPU  RESIDENT  LOOP/TAG  TYPES





By using certain special purpose Loop Table and Data Table Array types, the user can create any of three different types of VPU resident loops.  The choices are: Supervisory Controller; Data Entry (ie. lab entry); or Calculated Value (non enterable).  Further, each of the three types can optionally be configured to support the VPU trending feature.  The user would need to decide prior to creating the loop what his needs are.





As illustrated, the structure of the VPU resident loop type is quite similar to the Unitec loop type, the major difference being the replacement of the Unitec devices by different elements.  The overall encapsulating of several elements is still apparent, but there is now a "VPU BLOC Calculation" taking the place of the Unitec I/O device and the trend device has been replaced by a "Trend Buffer Array".  Also notice that the "BLOC" is not pointed to by the Loop Table.  It's purpose is to transfer it's own version of the feedback into the Process Value and Trend Buffer locations using calculated or other non-Unitec data.   The trend buffer allows the trend displays to access and display this loops trend data in the same format as Unitec loops.  The Loop and Data Tables provide the same transparency for tagline and faceplate display.  





��
Just as with a Unitec loop, the VPU Supervisory loop stores it's SetPoint (SP) and Process Value (PV) in the Loop Data Table, description and units text in the Loop Table, and the loop statuses (Bad Rdg, Auto, Casc mode) in the Status Ordinal Array.  Also like Unitec loops is the full complement of merge symbols, appropriate to the loop type.





The different loop types are selected by specifying the appropriate "Template" (sometimes called "Skeleton") in BLOC builder.  The user may skip this level of detail by using the pre-defined audit trail macros instead (shown in the next section).  If one wishes to refer to the templates (in BLOC builder), the template names for the different VPU looptable types are:





  LOOP  TYPE�
WITH TREND SUPPORT�
WITHOUT TREND SUPPORT�
USES  LOOP DATA TABLE �
�
CONTROLLER�
VPUSPRTRDTBL�
VPUSPRTBL�
VPUSPRDTTBL�
�
CALCULATED VALUE�
VPUCALTRDTBL�
VPUCALTBL�
VPUCALDTTBL�
�
DATA (LAB) ENTRY �
DATENTTRDTBL�
DATENTTBL�
DATENTDTTBL�
�



If using trend types, the array for the trend buffer ("ACCMTRARY") must be added before the loop data table (since the loop table refers to it), and all arrays must be added before the algorithms can refer to them.





Audit Trail Macros to Create VPU Loops - Data Entry Loop





The audit trail macro shown below will create the simple data entry loop type without a trend.  (It does not seem useful to trend data entries.)


		VPU_DATENT.ATR


.M


.P "P1,Enter LPN (# only, ie. 1 for LPN1)"


.P "P2,Enter BTI (suffix only, ie. 1 for BTID1)"


.P "P3,Enter Tagname (ie. CAUS.1405, 9 char MAX)"


.P "P4,Enter Full Scale (ie. 100.0)"


.P "P5,Enter Zero Scale (ie. 0.0)"


.P "P6,Enter 18 character description"


.P "P7,Enter 5 character units"





! Audit Trail To Add a Data Entry VPU Loop Type





ADD ARRAY LPN'%P1'\BTID'%P2'\'%P3'dt,datentdttbl,,'%P4','%P5'





ADD ARRAY LPN'%P1'\BTIDP'%P2'\'%P3',datenttbl,,-


dummygroup_g,100,NULL,NULL,-


(LPN'%P1'\BTID'%P2'\'%P3'dt,Y,),"'%P6'","'%P3'","'%P7'"





MERGE SYM  LPN'%P1'\BTID'%P2'\'%P3'CV   LPN'%P1'\BTID'%P2'\'%P3'DT\ENTRYVALUE


MERGE SYM  LPN'%P1'\BTID'%P2'\'%P3'CUL  LPN'%P1'\BTID'%P2'\'%P3'DT\ENTRY_MAX


MERGE SYM  LPN'%P1'\BTID'%P2'\'%P3'CLL  LPN'%P1'\BTID'%P2'\'%P3'DT\ENTRY_MIN


�
Audit Trail Macros to Create VPU Loops - Supervisory Control Loop





The audit trail macro shown below creates a supervisory loop which includes trending   The rates of accumulation are prompted and entered into the algorithm.  As with the Unitec, the fast trend is in units of 1/4 Sec while the slow is 5 Sec.  The prompts for symbol name for feedback and ordinal for bad status must already be defined prior to running the macro.





		VPU_SUPCTL.ATR


.M


.P "P1,Enter LPN (# only, ie. 1 for LPN1)"


.P "P2,Enter BTI (suffix only, ie. 1 for BTID1)"


.P "P3,Enter Tagname (ie. CAUS.1405, 9 char MAX)"


.P "P4,Enter Full Scale (ie. 100.0)"


.P "P5,Enter Zero Scale (ie. 0.0)"


.P "P6,Enter 18 character description"


.P "P7,Enter 5 character units"


.P "P8,Enter Defined Symbol Name for Feedback (ie. ICP29D1CV)"


.P "P9,Enter Ordinal Symbol Name for Bad Status (ie. ICP29D1SOA(1) )"


.P "P10,Enter Fast Trend Rate in 1/4 Sec Units"


.P "P11,Enter Slow Trend Rate in 5 Sec Units"





! Audit Trail To Add a Data Entry Vpu Loop Type





ADD ARRAY LPN'%P1'\BTID'%P2'\'%P3'tar,accmtrary,"ACCUM TREND ARR FOR '%P3'"


ADD ARRAY LPN'%P1'\BTID'%P2'\'%P3'dt,vpusprdttbl,,'%P4','%P5','%P4','%P5'





ADD ARRAY LPN'%P1'\BTIDP'%P2'\'%P3',vpusprtrdtbl,,-


dummygroup_g,100,NULL,NULL,(LPN'%P1'\BTID'%P2'\'%P3'soa,Y,),-


LPN'%P1'\BTID'%P2'\'%P3'dt,0,"'%P6'","'%P3'","'%P7'",-


LPN'%P1'\BTID'%P2'\'%P3'tar





MERGE SYM LPN'%P1'\BTIDP'%P2'\'%P3'TAG  LPN'%P1'\BTIDP'%P2'\'%P3'\TAGNAME


MERGE SYM LPN'%P1'\BTIDP'%P2'\'%P3'LOP  LPN'%P1'\BTIDP'%P2'\'%P3'\LOOPNAME


MERGE SYM LPN'%P1'\BTIDP'%P2'\'%P3'UNT  LPN'%P1'\BTIDP'%P2'\'%P3'\PRCSUNIT


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'PV    LPN'%P1'\BTID'%P2'\'%P3'DT\PV


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SP    LPN'%P1'\BTID'%P2'\'%P3'DT\SP


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SPU   LPN'%P1'\BTID'%P2'\'%P3'DT\SETPOINTMX


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SPL   LPN'%P1'\BTID'%P2'\'%P3'DT\SETPOINTMN


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SUL   LPN'%P1'\BTID'%P2'\'%P3'DT\DISPLAYMX


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SLL   LPN'%P1'\BTID'%P2'\'%P3'DT\DISPLAYMN


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SRQ   LPN'%P1'\BTID'%P2'\'%P3'SOA\2


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SOC   LPN'%P1'\BTID'%P2'\'%P3'SOA\3


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'SRY   LPN'%P1'\BTID'%P2'\'%P3'SOA\4


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'TUN   LPN'%P1'\BTID'%P2'\'%P3'SOA\5


MERGE SYM LPN'%P1'\BTID'%P2'\'%P3'CRQ   LPN'%P1'\BTID'%P2'\'%P3'SOA\6





ADD SYM '%P3'SSL='%P4',(VAR,REAL),LPN'%P1'\BTID'%P2',"Sum Of Limits"





SET DEF ERR CONT


ADD BCT bc_2sec,CHAINRES


ADD EVENT RATE,2,1,SEC,MEMRES,LPN'%P1'\ev_2sec,bc_2sec,100


ADD CHAIN ch_2sec,,bc_2sec,btib1,1,1,1,(COLD,TEPID),btid1,lpn1\btid1\$l.on


SET DEF ERR STOP





ADD BLOCK lpn1\ch_2sec\bl_'%P3',,1,lpn1\btid1\$l.on,ACTIVE





ADD ALG  lpn1\ch_2sec\bl_'%P3',,MPL


A


! Following code will load value from '%P8' into loop feedback '%P3'PV


! and the bad status ordinal from '%P9' into status ord '%P3'soa\1


'%P3'PV = '%P8';


'%P3'SOA(1) = '%P9';


END;


.


X





! This load alg is used to load the trended value into working storage.





ADD ALG  lpn1\ch_2sec\bl_'%P3',MPL,LOAD,'%P3'PV





! Then the ACCMTREND algorithm puts the '%P3'PV variable into the array.


ADD ALG  lpn1\ch_2sec\bl_'%P3',LOAD,ACCMTREND,'%P10','%P11',-


LPN'%P1'\BTID'%P2'\'%P3'sul,-


LPN'%P1'\BTID'%P2'\'%P3'sll,-


LPN'%P1'\BTID'%P2'\'%P3'tar





The execution of the macro in interactive mode is shown below.


(NOTE: remember to ignore the 130 character limit suggested by the prompt)


BLOC: @VPU_SUPCTL.ATR


"P1,Enter LPN (# only, ie. 1 for LPN1)"


Enter data/130 chars: 1





"P2,Enter BTI (suffix only, ie. 1 for BTID1)"


Enter data/130 chars: 1





"P3,Enter Tagname (ie. CAUS.1405, 9 char MAX)"


Enter data/130 chars: CAUS.1411





"P4,Enter Full Scale (ie. 100.0)"


Enter data/130 chars: 100.





"P5,Enter Zero Scale (ie. 0.0)"


Enter data/130 chars: 0.





"P6,Enter 18 character description"


Enter data/130 chars: Caust Chrg TK1411





"P7,Enter 5 character units"


Enter data/130 chars: # Chg





"P8,Enter Defined Symbol Name for Feedback (ie. ICP29D1CV)"


Enter data/130 chars: ICP89D1CV





"P9,Enter Ordinal Symbol Name for Bad Status (ie. ICP29D1SOA(1) )"


Enter data/130 chars: ICP89D1SOA(1)





"P10,Enter Fast Trend Rate in 1/4 Sec Units"


Enter data/130 chars: 16





"P11,Enter Slow Trend Rate in 5 Sec Units"


Enter data/130 chars: 0





! Audit Trail To Add a Data Entry Vpu Loop Type


ADD ARRAY LPN1\BTID1\CAUS.1411tar,accmtrary,"ACCUM TREND ARR FOR CAUS.1411"


ADD ARRAY LPN1\BTID1\CAUS.1411dt,vpusprdttbl,,100.,0.,100.,0.





ADD ARRAY LPN1\BTIDP1\CAUS.1411,vpusprtrdtbl,,-


dummygroup_g,100,NULL,NULL,(LPN1\BTID1\CAUS.1411soa,Y,),-


LPN1\BTID1\CAUS.1411dt,0,"CAUST CHRG TK1411","CAUS.1411","# CHG",-


LPN1\BTID1\CAUS.1411tar





MERGE SYM LPN1\BTIDP1\CAUS.1411TAG  LPN1\BTIDP1\CAUS.1411\TAGNAME


MERGE SYM LPN1\BTIDP1\CAUS.1411LOP  LPN1\BTIDP1\CAUS.1411\LOOPNAME


MERGE SYM LPN1\BTIDP1\CAUS.1411UNT  LPN1\BTIDP1\CAUS.1411\PRCSUNIT


MERGE SYM LPN1\BTID1\CAUS.1411PV    LPN1\BTID1\CAUS.1411DT\PV


MERGE SYM LPN1\BTID1\CAUS.1411SP    LPN1\BTID1\CAUS.1411DT\SP


MERGE SYM LPN1\BTID1\CAUS.1411SPU   LPN1\BTID1\CAUS.1411DT\SETPOINTMX


MERGE SYM LPN1\BTID1\CAUS.1411SPL   LPN1\BTID1\CAUS.1411DT\SETPOINTMN


MERGE SYM LPN1\BTID1\CAUS.1411SUL   LPN1\BTID1\CAUS.1411DT\DISPLAYMX


MERGE SYM LPN1\BTID1\CAUS.1411SLL   LPN1\BTID1\CAUS.1411DT\DISPLAYMN


MERGE SYM LPN1\BTID1\CAUS.1411SRQ   LPN1\BTID1\CAUS.1411SOA\2


MERGE SYM LPN1\BTID1\CAUS.1411SOC   LPN1\BTID1\CAUS.1411SOA\3


MERGE SYM LPN1\BTID1\CAUS.1411SRY   LPN1\BTID1\CAUS.1411SOA\4


MERGE SYM LPN1\BTID1\CAUS.1411TUN   LPN1\BTID1\CAUS.1411SOA\5


MERGE SYM LPN1\BTID1\CAUS.1411CRQ   LPN1\BTID1\CAUS.1411SOA\6


ADD SYM CAUS.1411SSL=100.,(VAR,REAL),LPN1\BTID1,"Sum Of Limits"





SET DEF ERR CONT


ADD BCT bc_2sec,CHAINRES


ADD EVENT RATE,2,1,SEC,MEMRES,LPN1\ev_2sec,bc_2sec,100


ADD CHAIN ch_2sec,,bc_2sec,btib1,1,1,1,(COLD,TEPID),btid1,lpn1\btid1\$l.on


SET DEF ERR STOP





ADD BLOCK lpn1\ch_2sec\bl_CAUS.1411,,1,lpn1\btid1\$l.on,ACTIVE





ADD ALG  lpn1\ch_2sec\bl_CAUS.1411,,MPL


! Following code will load value from ICP89D1CV into loop feedback CAUS.1411PV


! and the bad status ordinal from ICP89D1soa(1) into status ord CAUS.1411SOA\1


CAUS.1411PV = ICP89D1CV


CAUS.1411SOA(1) = ICP89D1soa(1);


END;


.


X


! This load alg is used to load the trended value into working storage.





ADD ALG  lpn1\ch_2sec\bl_CAUS.1411,MPL,LOAD,CAUS.1411PV





! Then the ACCMTREND algorithm puts the CAUS.1411PV variable into the array.


ADD ALG  lpn1\ch_2sec\bl_CAUS.1411,LOAD,ACCMTREND,16,0,-


LPN1\BTID1\CAUS.1411sul,-


LPN1\BTID1\CAUS.1411sll,-


LPN1\BTID1\CAUS.1411tar





The resulting code uses an entry of 16 for a 4 Second collection rate (16/4).  If the entry of either rate is not an even multiple of the block rate, the time intervals will be "pro-rated" so that the next block execution will accumulate an overdue point.  A consideration is that the time intervals for collection are only changeable via BLOC builder, rather than through the Reconfiguration displays as is the case with Unitec trends.








CODE FOR CHARGE METER EXAMPLE





The charge meter logic is created by the macro shown below.  The purpose for including it is to demonstrate the use of the status ordinals to provide status display and recognize mode change commands on the tagline.  In the following listing, you see that the code monitors the ordinal 'tag'SRQ as the "start charge" initiator.  This ordinal is set by any pressing of the "Start/Auto" button with this tag is selected in the tagline.  Likewise, the ordinals 'tag'SOC and 'tag'CRQ make the loop appear in 'AUTO' and 'CASC' respectively.  The code is primarily the not ready, auto mode, and cascade mode logic along with the two Stepfn algorithms at the end.  The Stepfn algorithms control the transmission of status and numeric data (Chrg Tgt and zero accumulation) at the begging and end of a charge.  While the charge is in progress, the logic monitors for failure or other not ready conditions, to initiate charge shutdown.





In our example, the feedback is a flow integration (summation) being performed on an ICP board, which does not have Real Time Trending capability.  As a footnote, the trending requirement could have been done using line graphs using the datalogger/historian, but  it was important to have the compatibility with current trending and also support setpoint entry from the tagline.





.P "P1, Enter the Charge TAGNAME (ie. CAUS_1411):"


.P "P2, Enter the DOP board FLAG ID name (ie. DOP11F01):"





  !=========================================


  ! DOP FLAG MERGE SYMBOLS


   add sym '%P1'.ST,'%P2'FIO\flag_1,"IOVAR FOR CHARGE START"


   mer sym '%P1'_ST,'%P2'SOA\18    ! FLAG FOR STARTED/IN PROGRESS





   add sym '%P1'.RS,'%P2'FIO\flag_2,"IOVAR FOR RE-START, DONT ZERO"


   mer sym '%P1'_RS,'%P2'SOA\19    ! FLAG FOR RE-START, DONT ZERO





   add sym '%P1'.FN,'%P2'FIO\flag_3,"IOVAR FOR CHARGE FINISHED"


   mer sym '%P1'_FN,'%P2'SOA\20    ! FLAG FOR CHARGE FINISHED





   add sym '%P1'.CL,'%P2'FIO\flag_4,"IOVAR FOR VALVE OPEN-CHARGING"


   mer sym '%P1'_CL,'%P2'SOA\21    ! FLAG FOR VALVE OPEN-CHARGING





   add sym '%P1'.FA,'%P2'FIO\flag_5,"IOVAR FOR CHARGE FAILURE"


   mer sym '%P1'_FA,'%P2'SOA\22    ! FLAG FOR CHARGE FAILURE





  !=========================================


ADD SYM '%P1'STP,OFF,ORD,lpn1\btid1


ADD SYM '%P1'ALM,OFF,ORD,lpn1\btid1





�
ADD BLOCK lpn1\ch_2sec\b.'%P1',,4,lpn1\btid1\$l.on,ACTIVE





ADD ALGORITHM lpn1\chn1sec\b.'%P1',,mpl


 A


  !=========================================


  ! SET READY WHEN HEADER VALVES CLOSED 


  ! OR IF ALREADY CHARGING AND NOT FAILED


  !=========================================





'%P1'SRY = (NOT '%P1'_ST  AND '%P1'_CL)  


        OR ('%P1'SOC AND  NOT  '%P1'SOA(1)  


       AND  NOT '%P1'_FA);





  !=========================================


  ! IF NOT READY, CLEAR REQUEST - SO ANY 


  ! FAILURE OR PRIOR REQUEST CLEARS REQ


  ! OR WHEN FINISHED FLAG GOES TRUE AND 


  ! CURRENTLY CHARGING, CLEAR REQ


  !=========================================





IF NOT '%P1'SRY OR ('%P1'_FN AND '%P1'SOC) ;


   CLEAR '%P1'SRQ;


ENDIF;





  !=========================================


  ! IF CURRENTLY CHARGING BUT REQUESTED


  ! HAS CLEARED, THEN SET "CHARGE STOP" FLAG


  !=========================================





'%P1'STP = NOT '%P1'SRQ AND '%P1'SOC;





  !=========================================


  ! TUN IS USED FOR START CHARGE TRIGGER ORD FOR STEPFN


  ! = IF REQ AND RDY AND NOT CURRENTLY ON (SOC)


  !=========================================





'%P1'TUN = '%P1'SRY


        AND '%P1'SRQ


        AND  NOT '%P1'SOC;





  !=========================================


  ! USE CRQ TO INDICATE CHARGE IN PROGRESS


  !=========================================





'%P1'CRQ = '%P1'_ST


       AND  NOT '%P1'_FN


       AND '%P1'_CL;


			


  !=========================================


  ! generate and alarm if phase failure occurs


  ! while the charge in still in progress


  !=========================================





'%P1'ALM = '%P1'_FA AND '%P1'SOC;





END;


.


X





ADD ALGORITHM lpn1\chn1sec\b.'%P1',mpl,stepfn,-


lpn1\btid1\'%P1'TUN,-


"SENDFLOAT",lpn1\btid1\'%P1'SP,"IOVAR",lpn1\'%P1'tgt'%P2',"NO",-


"SENDFLAG",lpn1\btid1\$l.on,"IOVAR",lpn1\'%P1'.rs,"NO",-


"SENDFLAG",lpn1\btid1\$l.off,"IOVAR",lpn1\'%P1'.fa,"NO",-


"COMPARE","ON",lpn1\btid1\'%P1'_FN,lpn1\btid1\$t.instant,-


"COMPARE","OFF",lpn1\btid1\'%P1'_RS,lpn1\btid1\$t.instant,-


"SENDFLOAT",(lpn1\btid1\F0,Y,0.0),"IOVAR",lpn1\'%P1'qty'%P2',"NO",-


END,lpn1\btid1\'%P1'SOC,lpn1\btid1\'%P1'SOC





ADD ALGORITHM lpn1\chn1sec\b.'%P1',stepfn,stepfn,-


lpn1\btid1\'%P1'STP,-


"SENDFLAG",lpn1\btid1\$l.off,"IOVAR",lpn1\'%P1'.st,"NO",-


"ORD_ASSIGN",lpn1\btid1\'%P1'SOC,lpn1\btid1\$l.off,-


"SENDFLAG",lpn1\btid1\$l.off,"IOVAR",lpn1\'%P1'.fa,"NO",-


END,lpn1\btid1\$l.off,lpn1\btid1\$l.off








ADD ALGORITHM lpn1\chn1sec\b.'%P1',STEPFN\2,sendalrm0,-


'%P1'ALM,alarm0,lpn1\BTIDP2\'%P1',-


lpn1\BTIDP2\'%P1'\groupnmbr,(lpn1\BTID1\'%P1'sev,Y,250),-


(lpn1\BTID1\'%P1'lon,Y,"114-'%P2' '%P1' CHARGE FAILURE"),-


(lpn1\BTID1\'%P1'sho,Y,"'%P1'"),-


(lpn1\BTID1\'%P1'atg,Y,"'%P1'"),-


(lpn1\BTID1\'%P1'old,Y,0)








Summary





So we have available a method of configuring trends and tag accessible data table arrays that is quick and convenient.  You should be able to change the details of the macros presented here to suit your needs more specifically if desired.  Of course, as with most Venus presentations the macro & audit trail files are available on floppy disk for any interested parties..  It has proved useful in a variety of situations, I hope this will prove helpful for you as well.

